

Reactive Programming in Standard ML

Riccardo R. Pucella

Bell Laboratories

Lucent Technologies

600 Mountain Avenue

Murray Hill, NJ 07974 USA

riccardo@research.bell-labs.com

Abstract

Reactive systems are systems that maintain an ongoing interaction with their environment, activated by receiving input events from the environment and producing output events in response. Modern programming languages designed to program such systems use a paradigm based on the notions of instants and activations. We describe a library for Standard ML that provides basic primitives for programming reactive systems. The library is a low-level system upon which more sophisticated reactive behaviors can be built, which provides a convenient framework for prototyping extensions to existing reactive languages.

1 Introduction

We consider in this paper the problem of programming applications containing reactive subsystems. A reactive system is defined as a system that maintains an ongoing interaction with its environment [21], activated by receiving input events from the environment and producing output events in response. Typical examples of reactive systems are user interfaces, required to coordinate the various user requests (from the keyboard, the mouse and other devices) with information coming from the application (enabling or disabling input components and so on). Such systems generally decompose into independent parallel components cooperating to solve a given task, and exhibit a high degree of concurrency [16]. Because of this, programming reactive systems using traditional sequential languages can be difficult, and one often turns to concurrent languages to simplify the programming task.

In the past decade, a class of languages has emerged specifically for programming reactive systems, including imperative languages such as Esterel [5] and declarative languages such as Signal [20] and Lustre [12]. Those languages are directly based on the model of reactive systems

as being activated by input events and producing output events. Their approach to programming reactive systems, referred to as the *reactive paradigm*, is to divide the life of a reactive system into *instants*, which are the moments where the system reacts. They allow the programmer to write statements that depend on instants. For example, a program may wait for the third instant where a given event occurs, and so on. Instants provide a notion of *logical time* to which programs may refer. This is in contrast to languages providing a notion of absolute (or real) time, for example Ada [1] with its *delay* statement.

This approach to programming reactive systems via instants has an interesting consequence. Instants act as a *global* logical time for a program, and thus the end of instants provide a consistent configuration of the state of the program, where one can make decisions before the next activation. This in turns allows for a clean specification of *preemption*, whereby one reaction can abort another reaction executing in parallel [4].

This paper describes a library for the programming language Standard ML (SML) [24] that implements the essence of the reactive paradigm, as described by Boussinot [6]: the notions of instants and activations. It permits the definition of SML expressions that can be activated and that specify control points denoting the end of instants.

The original purpose of the library was to help develop a reactive interface language for connecting user interface components, independently of the underlying window management system. The library also has features that make it interesting in its own right. It is built from a very small set of primitives, simplifying the task of analyzing programs using the library. It provides a framework for reactivity that fits naturally with the mostly-applicative programming style of SML. It can be implemented without sizable extensions to the language (none if the implementation provides first-class continuations or a similar facility). More importantly, it provides an opportunity to study the interaction of reactive

primitives with features missing from most existing reactive languages: higher-order functions and recursion.

The library is intended as a low-level framework implementing basic reactive functionality upon which one may build more sophisticated machinery. It can be used to investigate and prototype extensions to existing higher-level reactive languages, extending for example Esterel with higher-order facilities. Compilation for higher-level reactive languages is non-trivial, and using the reactive library as a target language for compilation, one can rapidly prototype extensions. Once a useful extension has been identified, effort can be put into finding a compilation process that generates code as efficient as possible.

The paper is organized as follows: the next section describes the primitives implemented by the library; Section 3 provides an example of reactive code written using the library; Section 4 gives the operational semantics of the reactive primitives; Section 5 compares the library to existing reactive frameworks, and Section 6 concludes with a discussion of future work.

2 The reactive library

In this section, we describe the reactive library and provide simple examples. Primitives for creating and activating basic reactive expressions are given, along with combinators to create new reactive expressions by combining existing ones.

2.1 Basic reactive expressions

The central notion defined by the reactive library is that of a *reactive expression*. A reactive expression is fundamentally an SML expression that defines instants. The basic primitives are shown in Figure 1. The function `rexp` creates a reactive expression out of its argument (a `unit -> unit` function). Activating the reactive expression will evaluate the argument of `rexp` until a call to `stop`, which marks the end of the current instant. The next activation of the reactive expression will resume the evaluation from the last point where a `stop` was called, until either another `stop` is called or the evaluation terminates. As a simple example, consider the following:

```
val exp = rexp (fn () => (print "FIRST\n";
                        stop();
                        print "SECOND\n"))
```

This code defines a reactive expression `exp` that prints `FIRST` the first time it is activated, and `SECOND` the second time it is activated. After the second activation, the reactive expression is *terminated*.

To activate a reactive expression from SML code, one applies the function `react`, which returns `true` if the expression is terminated and `false` otherwise. Activating a

terminated expression has no effect. The function `dup` creates a copy of the reactive expression, with its current state. Here's a sample session with the above example:

```
- react (exp);
FIRST                                (* first instant *)
false : bool
- val copy = dup (exp);              (* make a copy *)
val copy = - : rexp
- react (exp);
SECOND                               (* second instant *)
true : bool                          (* rexp terminates *)
- react (exp);
true : bool
- react (copy);                      (* activate the copy *)
SECOND
true : bool
```

The function `reactT` repeatedly activates the reactive expression until it terminates.

A reactive expression can furthermore relinquish control to another reactive expression, via the function `activate`. When a reactive expression calls `activate` on a reactive expression e , it effectively behaves as e until e terminates, at which point the reactive expression continues evaluation. For example, activating the reactive expression

```
rexp (fn ()=> (activate (exp);
               print "DONE\n"))
```

activates `exp`, stopping when `exp` stops. Once `exp` terminates, evaluation of the reactive expression continues and `DONE` is printed before the reactive expression itself terminates.

2.2 Combinators

Reactive expressions with a more complex behavior are created via combinators, which take existing reactive expressions and produce new ones. Figure 2 presents the most important combinators implemented by the library.

The combinator `merge` takes two reactive expressions e_1 and e_2 and returns a new reactive expression e with the following behavior: when e is activated, it activates e_1 and then e_2 . The reactive expression e terminates when both e_1 and e_2 terminate. For example, consider the reactive expression:

```
merge (rexp (fn ()=> (print "1"; stop (); print "2")),
      rexp (fn ()=> (print "A"; stop (); print "B")))
```

This reactive expression will print `1A` at the first instant, and `2B` at the second, then terminate. Note that the order of activation of the branches of the `merge` is determined. This ensures deterministic evaluation of the reactive expression. Alternate orders of evaluation can be specified by defining micro-instants (see Section 2.4).

The combinator `rif` takes a boolean-valued function and two reactive expressions e_1 and e_2 , and returns a new reactive expression e with the following behavior: when e is activated, the boolean-valued function is evaluated, and depending on the resulting value, either e_1 or e_2 is activated.

val rexp	: (unit -> unit) -> rexp	(* create a basic reactive expression *)
val stop	: unit -> unit	(* stop execution of current instant *)
val react	: rexp -> bool	(* activate a reactive expression *)
val reactT	: rexp -> unit	(* activate until termination *)
val dup	: rexp -> rexp	(* duplicate reactive expression *)
val activate	: rexp -> unit	(* relinquish control to reactive expression *)

Figure 1. Reactive primitives

The reactive expression e terminates if the selected reactive expression terminates. Note that the boolean-valued function is evaluated at *every* instant.

The other combinators are defined in terms of basic reactive expressions, and the combinators `merge` and `rif`. Consider for example the combinator `loop`. It takes a reactive expression e and creates a new reactive expression with the following behavior: upon activation, it saves a copy of e at its current state, and activates a copy of that saved expression; if that copy terminates, a new copy of the saved expression is created and activated. We initially save a copy of e and work exclusively on that copy in order not to be affected by external activations of e by other reactive expressions. The behavior just described can be implemented as follows:

```
fun loop (e) = let val saved_e = dup (e)
                fun l () = (activate (dup(saved_e));
                           l ())
              in
                rexp l
              end
```

2.3 Preemption

Preemption refers to the possibility for one reactive expression to force the termination of another reactive expression executing in “parallel” [4], i.e. in another branch of a `merge`. This is achieved in our framework by the SML exception mechanism. The following example shows how one branch of a `merge` can force termination of the whole `merge` expression:

```
exception Abort
let val m_exp =
  merge (rexp (fn ()=> (print "FIRST\n";
                       stop ();
                       raise Abort)),
        loop (rexp (fn ()=> (print "SECOND\n";
                              stop ())))))
in
  rexp (fn ()=> (activate (m_exp) handle Abort => ()))
end
```

Since the second branch of the `merge` is a `loop`, it never terminates, and thus the `merge` would never terminate if a preemption was not performed by the first branch.

2.4 Micro-instants

It is sometimes necessary to consider a subdivision of the notion of instant, to provide for a finer level of control. The following primitives are used to manage those so-called *micro-instants*:

```
val suspend : unit -> unit
val close   : rexp -> rexp
```

Micro-instants are created by calling the function `suspend` instead of `stop`. A reactive expression that calls the function `suspend` is said to be *suspended*. A suspended reactive expression behaves just like a stopped one, unless it is wrapped with a `close` combinator. Upon activation, a reactive expression created via a `close` combinator will repeatedly activate the wrapped expression until it stops or terminates. Whereas the `suspend` function splits instants into micro-instants, the `close` combinator performs the dual operation of combining the micro-instants together into a single instant.

For example, consider the following reactive expression:

```
close (merge (rexp (fn ()=> (print "SUSPENDING ";
                           suspend();
                           print "1"; stop ();
                           print "2"))),
          rexp (fn ()=> (print "A"; stop ();
                       print "B"))))
```

This reactive expression will print `SUSPENDING A1` at the first instant, and `2B` at the second, then terminate. The `merge` combinator activates the first reactive expression, which suspends after printing `SUSPENDING`, and then activates the second reactive expression, printing `A`. At this point, the `merge` is suspended, because its first branch is. The `close` combinator forces the completion of the suspended reactive expressions, and thus the first branch resumes its activation, printing `1` and then stopping. The second branch is already stopped, so the `merge` and the `close` stop. At the next instant, `2B` is printed in the standard manner, and termination follows.

There is an implicit `close` wrapping the reactive expression to which `react` is applied. It is therefore impossible to witness micro-instants at the level of the application that uses the reactive expression.

val merge	: rexp * rexp -> rexp	(* parallel activation	*)
val rif	: (unit -> bool) * rexp * rexp -> rexp	(* conditional activation	*)
val halt	: unit -> rexp	(* simply stop at every instant	*)
val nothing	: unit -> rexp	(* terminate immediately	*)
val loop	: rexp -> rexp	(* reactivate rexp upon termination	*)
val terminate	: (unit -> bool) * rexp -> rexp	(* activate rexp if condition is false	*)
val init	: (unit -> unit) * rexp -> rexp	(* call fn before every activation of rexp	*)
val await	: (unit -> bool) * rexp -> rexp	(* block activation of rexp until true	*)
val when	: (unit -> bool) * rexp -> rexp	(* activate rexp when true else stop	*)
val repeat	: int * rexp -> rexp	(* reactivate rexp a 'n' times	*)

Figure 2. Reactive combinators

A common use for `suspend` and micro-instants is to suspend the activation of a reactive expression until information has been collected from the activation of other reactive expressions. For example, one can implement various broadcast communication mechanisms between reactive expressions using micro-instants [6, 10].

3 An example: A simple keypad controller

Let us now consider an example of reactive code in some detail. Interesting applications of the reactive approach are found in the programming of widgets for user interface toolkits, such as `eXene` [17]. Intuitively, a widget is an element of a user interface that encapsulates some behavior. Basic widgets include buttons and text editing fields, and menus. More complex widgets can be build up from other widgets, like dialog boxes and so on. The obvious advantage of encapsulating behavior inside widgets is that they can be reused in different applications.

Programming a new widget involves selecting the widgets it will be built from and then programming the controlling behavior of the widget (also know as the controller), taking into account the input from its sub-widgets and the expected interface of the widget. The key observation is that the controller is fundamentally a reactive system.

Suppose we want to construct a widget representing a numeric keypad made up of

- ten push buttons for the digits;
- a push button for `CLEAR`, that resets the number currently entered to 0;
- a push button for `ENTER`, that prints the number currently entered and then resets it to 0.

Suppose moreover that we parameterize the widget with respect to an integer n , representing the size of the number buffer (the number of digits it will accept).

Here is how we could represent the behavior of the keypad using the reactive library. We program the keypad controller as a reactive expression. We assume a mechanism

that waits for windowing system events (such as button presses) and activates the controller if an event concerns it. To access the state of the world, we assume we have boolean-valued functions

```
val kDigitPressed : unit -> bool
val kClearPressed : unit -> bool
val kEnterPressed : unit -> bool
```

that inform us if the corresponding button has been pressed in the current instant and a function

```
val kDigitValue : unit -> int
```

that returns the last digit pressed¹.

We define a reactive expression for every button of the keypad. The full controller will simply be a merge of all these reactive expressions. This approach simplifies the task of adding new buttons to the keypad. The code for the controller is given in Figure 3.

The reactive expression `enter_rexp` corresponding to the `ENTER` button is simple: At every instant, we check if `ENTER` has been pressed, if so, we print the number currently entered, and then clear it and terminate. Otherwise, we stop and wait for the next instant. The reactive expression `clear_rexp` handling `CLEAR` is the same, except that we do not print the number currently entered. The reactive expression `digit_rexp` handling a digit waits for a digit to be pressed, and computes the new number before terminating.

Recall that the keypad is parameterized by an integer n representing the size of the number buffer. We define a reactive expression `getnum_rexp` that will accumulate exactly n digits by activating `digit_rexp` exactly n times and then doing nothing for the subsequent instants after n digits have been pressed.

All of these expressions are gathered together to form the full controller, which is obtained by applying the function `mkController` to an integer representing the desired size

¹No usable user interface toolkit would require us to access the state of the interface via such functions. We simply abstract away from the problem of communicating windowing system events via this artificial interface.

```

fun mkController (n) =
  let exception Clear
      val num = ref (0)
      fun clear () = (num := 0; raise Clear)
      val enter_rexp = rif (kEnterPressed,
                           rexp (fn () => (print (Int.toString (!num));
                                           clear ())),
                           halt ())
      val clear_rexp = rif (kClearPressed, rexp (clear), halt ())
      val digit_rexp = rif (kDigitPressed,
                           rexp (fn () => (num := !num * 10 + kDigitValue ())),
                           halt ())
      val getnum_rexp = rexp (fn () => (activate (repeat (n, digit_rexp));
                                       activate (halt ())))
  in
    loop (rexp (fn () => activate (enter_rexp || clear_rexp || getnum_rexp)
                                handle Clear => ()))
  end

```

Figure 3. Definition of the keypad controller

of the number buffer. The core of the controller is the reactive expression that activates `enter_rexp`, `clear_rexp` and `getnum_rexp` concurrently². Since `getnum_rexp` never terminates, the merged expression never terminates by itself. However, the function `clear` is used to raise a `Clear` exception. When ENTER or CLEAR is pressed, the exception is raised and intercepted by the exception handler of the main reactive expression; the expression terminates and loops, awaiting for another button press.

Suppose we wish to add a NEG button to the keypad, that negates the number currently entered. We need only provide a function `kNegPressed`, and a new reactive expression to handle this case:

```

val neg_rexp = rif (kNegPressed,
                  rexp (fn () => (num := ~(!neg))),
                  halt ())

```

We can then add `neg_rexp` to the general merge of the keypad controller. It is also possible to parametrize the controller over the buttons and corresponding reactive expressions used to implement them.

4 Operational semantics

We describe the semantics of reactive expressions in terms of a simple functional language, in the spirit of [3, 27]. The syntax of the language is given by the following grammar:

$$\begin{aligned}
 M \quad = \quad & x \mid () \mid M_1 \ M_2 \mid (M_1, M_2) \mid (M_1; M_2) \\
 & \mid (\text{fn } x \Rightarrow M) \mid (\text{rec } f(x) \Rightarrow M)
 \end{aligned}$$

where x and f are alphabetic identifiers. The semantic objects are given in Figure 4. As in [3], the set of expressions is a superset of both the set of values and the set of lexical

²The operator `||` is simply an infix version of `merge`, that allows for a clearer presentation of nested merged reactive expressions.

phrases. The set of identifiers includes all possible alphabetic identifiers, including the constructors and basic values. The constructors and basic values define the reactive behavior of the language. They have no special syntax beyond their existence as identifiers. The notation $\xrightarrow{\text{fin}}$ is used to denote a finite mapping.

A reactive expression is tagged with a unique reactive ID: whenever a new reactive expression is created via `rexp` or by applying a combinator to existing reactive expressions, a new reactive ID is allocated and associated to the reactive expression. A reactive environment $R \ S$ is used to store information relating to reactive expressions. The map R holds the bindings between reactive IDs and actual reactive expressions, stored as constructed values. The map S stores the current state of reactive expressions, indexed by reactive ID. The state of a reactive expression is either stopped, suspended or terminated. Given M a finite map, we use the notation $M[m : v]$ to denote the new map defined by:

$$M[m : v](m') = \begin{cases} M(m') & \text{if } m' \neq m \\ v & \text{if } m' = m \end{cases}$$

The semantics is described using Plotkin's Structural Operational Semantics [26], and extends the semantics of Reactive C given in [7].

The semantics we describe has two levels. The first level is a semantics for the core language, given in Figure 5, expressed as a conventional reduction rule semantics. The semantics of the core language is complicated by the fact that expressions can occur in two contexts: the normal sequential context, and in the context of a reactive expression (captured by a `rexp` constructor). The reduction relation

$$e, R \ S \xrightarrow{\alpha} e', R' \ S'$$

denotes the reduction of expression e into expression e' , possibly transforming the reactive environment $R \ S$ into

r	$\in$	ReactiveId
x, y, z, f	$\in$	Identifier
$()$	$\in$	Unit = $\{()\}$
c	$\in$	Constructor = $\{\text{rexp}, \text{merge}, \text{rif}, \text{close}\}$
b	$\in$	Basic Value = $\{\text{stop}, \text{suspend}, \text{activate}, \text{dup}\}$
R	$\in$	ReactiveSet = ReactiveId $\xrightarrow{\text{fin}}$ ConstructedValue
s	$\in$	State = $\{\text{STOP}, \text{SUSP}, \text{END}\}$
S	$\in$	StateSet = ReactiveId $\xrightarrow{\text{fin}}$ State
v	$\in$	Value = Unit $\cup$ ReactiveId $\cup$ Constructor $\cup$ ConstructedValue $\cup$ BasicValue $\cup$ Closure $\cup$ ValPair
$\langle c, v \rangle$	$\in$	ConstructedValue = Constructor $\times$ Value
(v_1, v_2)	$\in$	ValPair = Value $\times$ Value
$(\text{fn } x \Rightarrow e)$	$\in$	Closure = Identifier $\times$ Expression
e	$\in$	Expression = Value $\cup$ Application $\cup$ Identifier $\cup$ ExpPair $\cup$ RecExp
$e_1 e_2$	$\in$	Application = Expression $\times$ Expression
(e_1, e_2)	$\in$	ExpPair = Expression $\times$ Expression
$(\text{rec } f(x) \Rightarrow e)$	$\in$	RecExp = Identifier $\times$ Expression $\times$ Expression
RS	$\in$	ReactiveEnvironment = ReactiveSet $\times$ StateSet

Figure 4. Semantic objects

$R' S'$. The reduction is labeled by an action α : **END** if the reduction terminates instantly, **STOP** if the reduction is stopped and **SUSP** if the reduction is suspended. Stopped and suspended reductions can only occur within the context of **rexp** constructed expression.

As we already noted, the only reductions allowed for the core language in the normal sequential context are terminated reductions (labeled **END**). The function **react** is the only function that may be called from the sequential context. We do not give the semantics of **react** to unclutter the presentation of the rules, but **react** behaves as **activate** in the sequential context — returning **true** or **false** depending on the resulting status of the reactive expression to which it is applied.

In the context of a reactive expression, the core language is allowed the full range of labeled reductions. The basic functions **stop**, **suspend**, **activate** and **reset** are allowed in such a context.

The interesting rules in that part of the semantics are the rules for **activate**, which acts upon reactive expressions (really reactive IDs representing reactive expressions). The intuitive interpretation of **activate** is to activate the given reactive expression, which propagates the activation according to the structure of the reactive expression.

The rules for **activate** involve the reduction relation

$$r, RS \xrightarrow{\alpha} R' S'$$

that act on a reactive expression whose ID is r . Again, the reduction is labeled by an action α indicating if the reduc-

tion is terminated, stopped or suspended. The intuition behind this reduction relation is that if the reactive expression denoted by r is stopped, then the activation is not propagated, and **activate** returns immediately. If the reaction is not terminated, then the activation is propagated to the reactive expression via the reduction $\xrightarrow{\alpha}$. The reduction relation $\xrightarrow{\alpha}$ is not formally necessary, but does greatly simplify the semantic rules.

The actual activation of reactive expressions is expressed by the reduction relation

$$r, RS \xRightarrow{\alpha} R' S'$$

denoting the activation of the reactive expression whose ID is r . The rule to apply depends on the structure of the reactive expression, whose constructed value is extracted from the reactive environment. Again, this reduction rule is labeled by an action α indicating if the reaction terminates, stops or suspends. Figure 6 gives the semantics of reactive expressions. If the reactive expression is a **rexp** constructed value, the reduction of the expression involves the reduction of an expression in the core language via $\xrightarrow{\alpha}$ transitions.

The semantics of the **merge** combinator use a function $\star$ on actions, defined as follows:

$\star$	SUSP	STOP	END
SUSP	SUSP	SUSP	SUSP
STOP	SUSP	STOP	STOP
END	SUSP	STOP	END

$$\begin{array}{c}
\frac{e_1, R S \xrightarrow{\alpha} e'_1, R' S'}{e_1 e_2, R S \xrightarrow{\alpha} e'_1 e_2, R' S'} \quad \frac{e, R S \xrightarrow{\alpha} e', R' S'}{v e, R S \xrightarrow{\alpha} v e', R' S'} \\
\\
\frac{e_1, R S \xrightarrow{\alpha} e'_1, R' S'}{(e_1, e_2), R S \xrightarrow{\alpha} (e'_1, e_2), R' S'} \quad \frac{e, R S \xrightarrow{\alpha} e', R' S'}{(v, e), R S \xrightarrow{\alpha} (v, e'), R' S'} \\
\\
\frac{e_1, R S \xrightarrow{\alpha} e'_1, R' S'}{(e_1; e_2), R S \xrightarrow{\alpha} (e'_1; e_2), R' S'} \quad \frac{e_1, R S \xrightarrow{\alpha} v, R' S'}{(e_1; e_2), R S \xrightarrow{\alpha} e_2, R' S'} \\
\\
\frac{}{(\text{fn } x \Rightarrow e) v, R S \xrightarrow{\text{END}} e\{v/x\}, R S} \\
\\
\frac{}{(\text{rec } f(x) \Rightarrow e), R S \xrightarrow{\text{END}} (\text{fn } x \Rightarrow e\{(\text{rec } f(x) \Rightarrow e)/f\}), R S} \\
\\
\frac{r \notin \text{dom}(R)}{c v, R S \xrightarrow{\text{END}} r, R[r : \langle c, v \rangle] S[r : \text{STOP}]} \\
\\
\frac{r, R S \xrightarrow{\text{END}} R' S'}{\text{activate}(r), R S \xrightarrow{\text{END}} (), R' S'} \quad \frac{r, R S \xrightarrow{\alpha} R' S' \quad \alpha \neq \text{END}}{\text{activate}(r), R S \xrightarrow{\alpha} \text{activate}(r), R' S'} \\
\\
\frac{}{\text{stop}(), R S \xrightarrow{\text{STOP}} (), R S} \quad \frac{}{\text{suspend}(), R S \xrightarrow{\text{SUSP}} (), R S} \\
\\
\frac{R(r) = \langle c, v \rangle \quad S(r) = s \quad r' \notin R}{\text{dup}(r), R S \xrightarrow{\text{END}} r', R[r' : \langle c, v \rangle] S[r' : s]} \\
\\
\frac{S(r) = \text{END}}{r, R S \xrightarrow{\text{END}} R S} \quad \frac{S(r) \neq \text{END} \quad r, R S \xrightarrow{\alpha} R' S'}{r, R S \xrightarrow{\alpha} R' S'}
\end{array}$$

Figure 5. Semantics of core language

4.1 Implementation

The implementation of the library is a direct translation of the operational semantics. The core of the implementation is a function `step` that plays the role of the $\Rightarrow$ transition in the semantics. It is used to activate a reactive expression and returns the state of the expression after the activation. Every basic combinator may be expressed by the `step` function and the basic reactive expression constructor, by simply defining it via its semantic reduction rules. We therefore only need to concentrate on the implementation of `step`, `stop` and `suspend`.

The library was implemented with the Standard ML of New Jersey compiler [2]. The compiler provides `callcc` [22], an extension to SML that allows the expression of powerful control abstraction in a typed setting. It lets one grab the current continuation of the evaluation of an expression as a first-class object and resume it at will.

A reactive expression is implemented as a tuple containing the continuation of the reactive expression and the current state of the expression. Calling `step` on the tuple simply throws the stored continuation to resume the evaluation

of the expression, after saving the current continuation. This latter continuation will be thrown if `stop` is called from the reactive expression code. The function `stop` saves the current continuation of the reactive expression in the tuple, and throws the continuation saved by the `step` function, resuming the evaluation of the code calling `step`. The function `suspend` is similarly implemented. The technique used is analogous to the one used by Wand [29] and Reppy [27] to implement concurrent threads via continuations.

5 Comparison with other reactive frameworks

The library described in this paper evolved from a desire to port the reactive framework of Reactive C [6] to the higher-order language SML. It is instructive to compare our system against both the original Reactive C and its derivative, the Java toolkit SugarCubes [11]. We also compare the library to various other frameworks for programming reactive systems.

$$\begin{array}{c}
\frac{R(r) = \langle \text{rexpr}, v_1 \rangle \quad v_1(), R S \xrightarrow{\alpha} e, R' S' \quad \alpha \neq \text{END}}{r, R S \xRightarrow{\alpha} R'[r : \langle \text{rexpr}, (\text{fn } () \Rightarrow e) \rangle] S'[r : \alpha]} \\
\\
\frac{R(r) = \langle \text{rexpr}, v_1 \rangle \quad v_1(), R S \xrightarrow{\text{END}} v, R' S'}{r, R S \xRightarrow{\text{END}} R' S'[r : \text{END}]} \\
\\
\frac{R(r) = \langle \text{merge}, (r_1, r_2) \rangle \quad S(r) \neq \text{SUSP} \quad r_1, R S \xRightarrow{\alpha_1} R' S' \quad r_2, R' S' \xRightarrow{\alpha_2} R'' S''}{r, R S \xRightarrow{\alpha_1 * \alpha_2} R'' S''[r : \alpha_1 * \alpha_2]} \\
\\
\frac{R(r) = \langle \text{merge}, (r_1, r_2) \rangle \quad S(r_1) = S(r_2) = \text{SUSP} \quad r_1, R S \xRightarrow{\alpha_1} R' S' \quad r_2, R' S' \xRightarrow{\alpha_2} R'' S''}{r, R S \xRightarrow{\alpha_1 * \alpha_2} R'' S''[r : \alpha_1 * \alpha_2]} \\
\\
\frac{R(r) = \langle \text{merge}, (r_1, r_2) \rangle \quad S(r_1) = \text{SUSP} \quad S(r_2) \neq \text{SUSP} \quad r_1, R S \xRightarrow{\alpha} R' S'}{r, R S \xRightarrow{\alpha * S(r_2)} R' S'[r : \alpha * S(r_2)]} \\
\\
\frac{R(r) = \langle \text{merge}, (r_1, r_2) \rangle \quad S(r_1) \neq \text{SUSP} \quad S(r_2) = \text{SUSP} \quad r_2, R S \xRightarrow{\alpha} R' S'}{r, R S \xRightarrow{S(r_1) * \alpha} R' S'[r : S(r_1) * \alpha]} \\
\\
\frac{R(r) = \langle \text{rif}, (v, r_1, r_2) \rangle \quad S(r) \neq \text{SUSP} \quad v(), R S \xrightarrow{\text{END}} \text{true}, R' S' \quad r_1, R' S' \xRightarrow{\alpha} R'' S''}{r, R S \xRightarrow{\alpha} R'' S''[r : \alpha]} \\
\\
\frac{R(r) = \langle \text{rif}, (v, r_1, r_2) \rangle \quad S(r) \neq \text{SUSP} \quad v(), R S \xrightarrow{\text{END}} \text{false}, R' S' \quad r_2, R' S' \xRightarrow{\alpha} R'' S''}{r, R S \xRightarrow{\alpha} R'' S''[r : \alpha]} \\
\\
\frac{R(r) = \langle \text{rif}, (v, r_1, -) \rangle \quad S(r_1) = \text{SUSP} \quad r_1, R S \xRightarrow{\alpha} R' S'}{r, R S \xRightarrow{\alpha} R' S'[r : \alpha]} \\
\\
\frac{R(r) = \langle \text{rif}, (v, -, r_2) \rangle \quad S(r_2) = \text{SUSP} \quad r_2, R S \xRightarrow{\alpha} R' S'}{r, R S \xRightarrow{\alpha} R' S'} \\
\\
\frac{R(r) = \langle \text{close}, r' \rangle \quad r', R S \xRightarrow{\alpha} R' S' \quad \alpha \neq \text{SUSP}}{r, R S \xRightarrow{\alpha} R' S'[r : \alpha]} \\
\\
\frac{R(r) = \langle \text{close}, r' \rangle \quad r', R S \xRightarrow{\text{SUSP}} R' S' \quad r, R' S' \xRightarrow{\alpha} R'' S''}{r, R S \xRightarrow{\alpha} R'' S''}
\end{array}$$

Figure 6. Semantics of reactive expressions

5.1 Reactive C and SugarCubes

The principal difference between the formalism in this paper and the formalisms of both Reactive C and SugarCubes relates to the programming paradigm embodied by the underlying languages. The Reactive C formalism extends the imperative language C [23] where programs are viewed as sequence of commands. The formalism defines a “machine” executing a sequence of reactive “instructions”. The SugarCubes toolkit extends the object-oriented language Java [18], and also uses the same imperative approach.

To illustrate the differences between “reactive instructions” and reactive expressions as we defined them in this paper, observe that our framework can be expressed as a

datatype. Following SML’s notation, one may define the following:

```

datatype rexp = REXP of unit -> unit
              | MERGE of rexp * rexp
              | RIF of (unit -> unit) * rexp * rexp
              | ...

```

A reactive expression becomes a tree-shaped data structure, and `react` simply walks the given tree. In fact, the semantics of reactive expressions given in Section 4 uses exactly this view of reactive expressions as a constructed datatype. In this framework, the leaves of the structure are basic reactive expressions that contain arbitrary `stop`, `suspend` and `activate` calls. Sample reactive code would look like:

```

MERGE (REXP (fn () => (print "1"; stop(); print "2")),
      REXP (fn () => (print "A"; stop(); print "B")))

```


If we were to implement the reactive “instructions” approach in SML via a datatype description as above, we would obtain something like the following:

```
datatype rinst = EXP of unit -> unit
| STOP
| SUSPEND
| ACTIVATE of rinst
| SEQUENCE of rinst list
| MERGE of rinst
| RIF of (unit -> unit) * rinst * rinst
| ...
```

We do not allow arbitrary calls to `stop` and `suspend` in basic expressions. Rather, the end of instants are explicitly specified in the datatype. Basic expressions always terminate immediately. This means that much of the structure that in our framework would fit in a basic reactive expression REXP now needs to be explicitly added to the datatype (for example, a way to describe sequences of reactive instructions). The sample code given above would now look like:

```
MERGE (SEQUENCE [EXP (fn ()=> print "1"), STOP,
  EXP (fn ()=> print "2")],
  SEQUENCE [EXP (fn ()=> print "A"), STOP,
    EXP (fn ()=> print "B")])
```

A reactive library implemented via reactive instructions (using the model of SugarCubes) is part of the Standard ML of New Jersey Library³.

The first approach, which we followed in this paper, allows for a clearer syntax, by directly using SML control-flow primitives (sequencing, local declarations) which need to be redefined in the datatype for the second approach. Moreover, the first approach allows one to easily reuse existing higher-order functions in a reactive way. One can easily write:

```
REXP (fn ()=> app (stop o print) ["1","2","3","4"] )
```

which prints one number of the list at every instant until termination. Expressing this reactive expression in the second approach seems difficult. On the other hand, the reactive code in the second approach is easier to analyze (for the purpose of compilation, for example), since the end of instants is fully characterized by the actual data structure representing the reactive code — there is no need to analyze the control-flow of an arbitrary SML expression calling `stop` and `suspend`.

5.2 Synchronous languages

Synchronous languages are among the most popular languages for programming reactive systems. These include Esterel [5], Lustre [12] and Signal [20]. These languages are all based on the same notions of instants and activations that we describe in this paper, but with important additions. In the case of Esterel, we have the following:

1. The instants are assumed to take zero time and are atomic. This is the *synchrony hypothesis*.
2. Communication between parallel reactions is done via broadcast signals, and is instantaneous.
3. Preemption can be triggered by the presence of a specified signal in the instant under consideration.

These characteristics allow the code for Esterel (and synchronous languages in general) to be efficiently compiled into a finite-state automaton, which can be translated into a program in a sequential language. There exists a translator taking the output of the Esterel compiler into SML code implementing the corresponding finite-state automaton⁴.

Our framework does not support the synchrony hypothesis of synchronous languages, and provides no communication mechanism between various parallel reactions beyond shared memory. As such, it does not support compilation into finite-state machines, and can be considered lower-level than synchronous languages. Boussinot and de Simone showed in [9] that it is possible to translate a synchronous language into a framework similar to the one described in this paper. This justifies our intended goal of using the library as a target language for experimental extensions to synchronous languages, such as higher-order synchronous languages. These extensions might not preserve finite-state semantics, but may still be useful as a convenient notation for various types of processes.

Higher-order extensions to synchronous languages include the work of Caspi and Pouzet [13] on extending the dataflow synchronous language Lustre with higher-order functions.

5.3 Fran

The Fran system [15] is a reactive framework for programming multimedia animations in Haskell [25]. It defines the notions of behaviors and events to program reactive animations. A behavior is fundamentally a function of time, and it is possible to specify behaviors with respect to events. The principal difference between the Fran approach and the one in this paper is that Fran is based on a continuous time model as opposed to our discrete time model divided into instants.

5.4 Coroutine facilities

Languages providing facilities for defining coroutines can be used to define a reactive framework such as the one we present in this paper. For example, the programming language Icon [19] provides *co-expressions*, which are expressions that can be suspended and resumed at a later time.

³J. Reppey, Personal communication, 1997.

⁴J. Riecke, Personal communication, 1997.

When it suspends, a co-expression needs to state to which other co-expression it is relinquishing control. Our basic notions of activation and instants can be viewed as a hierarchical use of co-expressions.

6 Future work

The interesting questions about the framework presented in this paper all relate to the interaction between the reactive formalism and the underlying mostly-functional approach of SML. For example, the current implementation of the `rif` combinator uses a `(unit -> unit)` function to be evaluated at every instant to determine which branch of the reactive conditional to activate. This means that any external value used by the test must be a reference.

Possible extensions to the framework include adding parameters to `react` that will be propagated through every combinator and used by `rif` during the evaluation of the conditional test. Dually, we can give a return type to reactive expressions, so that a value may be returned when a reaction stops or terminates. One should then augment the `merge` combinator with a function specifying how to combine the values returned by the two branches.

Further investigations into the interaction between reactivity and higher-order functions will involve the implementation of the reactive framework as a monad [28] in the purely functional language Haskell. Work by Claessen [14] on expressing concurrency as a monad via explicitly interleaved atomic actions closely follow our `merge` combinator. It would also be of interest to embed the reactive framework in a typed λ -calculus, in a way similar to the semantics of reactivity given in terms of a process calculus in [8].

7 Conclusion

We have described in this paper a reactive library for SML that implements the reactive paradigm exemplified by modern languages such as Esterel. The library provides primitives that capture the essence of the reactive paradigm, namely the notions of instants and activations. The library is intended to be a low-level system upon which more sophisticated reactive behavior can be built, providing a convenient framework for prototyping various higher-level reactive languages.

Acknowledgments

I wish to thank Frédéric Boussinot and John Reppy for careful readings of the manuscript, as well as Nevin Heintze, Hormoz Pirzadeh, Jon Riecke, Michael Sperber, Walid Taha and the anonymous referees for useful comments on various aspects of this work.

References

- [1] Ada 9X Mapping/Revision Team. *Ada 95 Reference Manual*. Intermetrics, Inc., 1995. ISO 8652:1995 (E).
- [2] A. W. Appel and D. B. MacQueen. A Standard ML compiler. In *Functional Programming Languages and Computer Architecture*, volume 274 of *Lecture Notes in Computer Science*, pages 301–324. Springer-Verlag, 1987.
- [3] D. Berry, R. Milner, and D. N. Turner. A semantics for ML concurrency primitives. In *Conference Record of the Nineteenth Annual ACM Symposium on Principles of Programming Languages*, pages 119–129. ACM Press, 1992.
- [4] G. Berry. Preemption in concurrent systems. In *Proceedings of FSTTCS'93*, volume 761 of *Lecture Notes in Computer Science*, pages 72–93. Springer-Verlag, 1993.
- [5] G. Berry and G. Gonthier. The Esterel synchronous programming language: Design, semantics, implementation. *Science of Computer Programming*, 19(2):87–152, 1992.
- [6] F. Boussinot. Reactive C: An extension of C to program reactive systems. *Software — Practice and Experience*, 21(4):401–428, 1991.
- [7] F. Boussinot. RC semantics using rewriting rules. Internal Report 18-92, ENSMP-CMA, 1992.
- [8] F. Boussinot. Reactivity and atomicity. Internal Report 20-93, ENSMP-CMA, 1993.
- [9] F. Boussinot and R. de Simone. The SL synchronous language. *IEEE Transactions on Software Engineering*, 22(4):256–266, 1996.
- [10] F. Boussinot and L. Hazard. Reactive scripts. In *Proceedings RTCSA'96*. IEEE, 1996.
- [11] F. Boussinot and J-F. Susini. The SugarCubes tool box — definition, April 1997. Available at <http://www.inria.fr/meije/rc/SugarCubes/>.
- [12] P. Caspi, D. Pilaud, N. Halbwachs, and J. Plaice. Lustre, a declarative language for programming synchronous systems. In *Conference Record of the Fourteenth Annual ACM Symposium on Principles of Programming Languages*. ACM Press, 1987.
- [13] P. Caspi and M. Pouzet. Synchronous Kahn networks. In *Proceedings of the 1996 ACM SIGPLAN International Conference on Functional Programming*. ACM Press, 1996.
- [14] K. Claessen. A poor man's concurrency monad. To appear in *Journal of Functional Programming*, 1997.
- [15] C. Elliott and P. Hudak. Functional reactive animation. In *Proceedings of the 1997 ACM SIGPLAN International Conference on Functional Programming*. ACM Press, 1997.
- [16] E. R. Gansner and J. H. Reppy. A foundation for user interface construction. In B. A. Myers, editor, *Languages for Developing User Interfaces*, chapter 14, pages 239–260. Jones and Bartlett Publishers, 1992.
- [17] E. R. Gansner and J. H. Reppy. A multi-threaded higher-order user interface toolkit. In Bass and Dewan, editors, *User Interface Software*, volume 1 of *Software Trends*, pages 61–80. John Wiley & Sons, 1993.
- [18] J. Gosling, B. Joy, and G. Steele. *The Java Language Specification*. Addison Wesley, 1996.
- [19] R. Griswold and M. Griswold. *The Icon Programming Language*. Peer-to-Peer Communications, Inc., third edition, 1996.
- [20] P. Le Guernic, A. Benveniste, P. Bournai, and T. Gautier. Signal: A data-flow oriented language for signal processing. *IEEE-ASSP*, 34(2):362–374, 1986.

- [21] D. Harel and A. Pnueli. On the development of reactive systems. In *Logic and Models of Concurrent Systems*, pages 477–498. Springer-Verlag, 1985.
- [22] R. Harper, B. F. Duba, and D. MacQueen. Typing first-class continuations in ML. *Journal of Functional Programming*, 3(4):465–484, 1993.
- [23] B. W. Kernighan and D. Ritchie. *The C Programming Language*. Prentice-Hall, 1988.
- [24] R. Milner, M. Tofte, R. Harper, and D. MacQueen. *The Definition of Standard ML (Revised)*. The MIT Press, Cambridge, Mass., 1997.
- [25] J. Peterson and K. Hammond. Report on the programming language Haskell, version 1.4. Technical report, Department of Computer Science, Yale University, 1997. Available from <http://www.haskell.org>.
- [26] G. D. Plotkin. A structural approach to operational semantics. Technical Report DAIMI FN-19, University of Aarhus, 1981.
- [27] J. H. Reppy. *Higher-Order Concurrency*. PhD thesis, Department of Computer Science, Cornell University, 1992. Available as Technical Report TR 92-1285.
- [28] P. Wadler. Monads for functional programming. In J. Jeuring and E. Meijer, editors, *Advanced Functional Programming*, volume 925 of *Lecture Notes in Computer Science*. Springer-Verlag, 1995.
- [29] M. Wand. Continuation-based multiprocessing. In *Conference Record of the 1980 Lisp Conference*, pages 19–28, 1980.